Paper 263-2011

ODS RTF: the Basics and Beyond

Lauren Haworth, Genentech, Inc., South San Francisco, CA

ABSTRACT

RTF is one of the best ODS destinations for sharing output with other users. All they need is a word processor to view your output. It also gives you and your users the ability to edit the output, add comments and other material, or combine the output with other reports.

This paper focuses on key techniques to get you up and running with RTF output, and some extra tricks you can use to impress the boss. It covers the basics: headers, page orientation, margins, page breaks, and page numbers. It also shows how to upgrade your RTF output with images, explanatory text, hyperlinks, and a table of contents. Finally, this paper shows how to use RTF control words and fields to implement advanced features.

INTRODUCTION

The paper uses Base SAS, and is aimed at beginning to intermediate ODS users. The examples are all taken from the book "ODS: the Basics and Beyond" (SAS Press, Fall 2009). All of the examples are written using SAS 9.2. You can find SAS 9.1.3 versions of the examples on the companion web site for the book.

ODS RTF TIP SHEET

There is no better set of reference guides to ODS than the new tip sheets that were developed by the Base SAS Team, as shown on the Base SAS R&D web site (http://support.sas.com/rnd/base/ods/index.html). There is a tip sheet that focuses specifically on ODS RTF. The tip sheets should not be considered a substitute for the ODS documentation, however, they are concise, accurate and extremely handy. And since they are maintained by the ODS developers, they represent the most up-to-date quick reference sheets available.

S.Sas. THE POWER TO KNOW.		Search support sas.com Search
support.sas.com	KNOWLEDGE BASE SUPPORT LEARNING CENTER CO	OMMUNITY
KNOWLEDGE BAS	E / FOCUS AREAS	Print E-mail Bookmark
FOCUS AREAS Base SAS ODTA Step ODS ODS DOCUMENT ODS HTML ODS MARKUP	Base SAS ODS (Output Delivery System) This page is part of the Base SAS Community and is an "anchor point" for lots of information regarding SAS ODS (Output Delivery System).	HOT TOPICS Tip Sheets (.pdf) ODS ODS DOCUMENT ODS PDF ODS RTF ODS TAGSETS.RTF (SAS 9.2)
- ODS PACKAGES - ODS PDF - ODS RTF - ODS & Microsoft Excel - SAS Notes for ODS	New and Notable • A SAS® Output Delivery System Menu for All Appetites and Applications (March 2009) • CSSSTYLE: Stylish Output with ODS and	PROC TEMPLATE tables PROC TEMPLATE styles Reporting PROCs styles Hash object Perl regular expressions

Figure 1: Tip Sheets

In addition to these handy tip sheets, on the R&D web site you will find links to papers by the developers and the most current information on R&D efforts in the Base SAS arena.

CREATING YOUR FIRST RTF FILE

The best thing about creating RTF output is that you have to learn only one new piece of syntax: just change the keyword HTML to RTF in your ODS statement. It takes only a single simple command to begin creating an RTF file:

```
ODS RTF FILE='myfile.rtf';
```

CHANGING THE STYLE

One simple change that you can make for RTF output is to use the STYLE= option in the invocation statement to generate a report with a completely different look and feel for the output.

```
ODS RTF FILE='myfile2.rtf' STYLE=JOURNAL;
```

Just as with the ODS HTML statement, you can list a literal filename, or use a fileref to refer to a previously defined filename. Next, you insert your procedure or DATA step to create some output. Finally, you close the RTF file with the following:

ODS RTF CLOSE;

To show how this works, we'll create two simple reports using PROC MEANS. For the first report, we'll use the default invocation. Then we'll create a second file using the JOURNAL style. Here is the code for the first report:

```
ODS RTF FILE='Payroll.rtf';
title 'Payroll Report';
proc means data=hr nonobs maxdec=0 mean sum;
class Department;
var AnnualSalary;
run;
ODS RTF CLOSE:
```

For the second file, the only change to the above program is to add the STYLE= option to the ODS RTF statement:

ODS RTF FILE='Payroll.rtf' STYLE=JOURNAL;

The resulting RTF files can be opened using your favorite word processor. Figure 2 shows what the two different styles look like if you open the output files with Microsoft Word.

Default	RTF Sty	JOURN	AL Styl	е		
Payro	oli Repor	Payroll Report				
The MEA	NS Proc	edure	The MEANS Procedure			
Analysis Variable :			Analysis Varia Anni	ible : Ann Ial Salary	ualSalary	
Analysis Variable : AnnualSalary Annual Salary		Department	Mean	Sui		
Department	Mean	Sum	Administration	51014	357100	
Administration	51014	3571000	Manufacturing	51507	1045600	
Manufacturing	51 507	10456000	Marketing	53465	384950	
Marketing	53465	3849500	R&D	53587	278650	
R&D	53 5 8 7	2786500				

Figure 2: Default and Journal RTF Styles

The RTF output looks similar to the HTML output, but the report table is more refined. The table headers are either shaded or bold, depending on the style used.

After the output is opened in a word processor, you can manipulate the output just like any other document. You can change the font face or size, use different colors or borders for the table, apply heading formats to the titles, or add explanatory text. In Word, you can also use the Table AutoFormat tool to create more elaborate table designs.

How is all this accomplished? The RTF file contains not just the information in your results, but also a variety of complex formatting information. Figure 3 shows a bit of the RTF file for this example, to give you an idea of how the files look internally.

{\rtfl\ansi\ansicpg1252\uc1\deff0\deflang1033\deflangfe1033 {\fonttbl {\f1\froman\fprq2\fcharset0 Times;} }{\colortbl;\red0\green0\blue0;\red0\green0\blue255;\red0\green255\blue255;\red0\green 255\blue0;\red255\green0\blue255;\red255\green0\blue0;\red255\green255\blue0;\red255\g reen255\blue255; \red0\green0\blue128;\red0\green128\blue128;\red0\green128\blue0;\red128\green0\blue12 8;\red128\green0\blue0;\red128\green128\blue0;\red128\green128\blue128;\red192\green19 2\blue192; }{\stylesheet{\widctlpar\adjustright\fs20\cgrid\snext0 Normal;}{*\cs10\additive **Default Paragraph Font;**} }{\info{\title V8 SAS System Output}{\author SAS Version 8}{\operator SAS Version 8}{\version1}{\creatim\yr2000\mo6\dy27\hr7\min23\sec37}} \widowctrl\ftnbj\aenddoc\formshade\viewkind1\viewscale75\pgbrdrhead\pgbrdrfoot\fet0\pa perw11519\paperh15023\marg1360\margr360\margt360\margb360 \sectd\linex0\endnhere\headery720\footery720\marglsxn360\margrsxn360\margtsxn360\margb sxn360 (Remaining RTF code trimmed to save space.)

Figure 3: Sample RTF Code

HEADERS AND FOOTERS IN WORD

The previous samples of output from Microsoft Word showed the document in print preview mode. This was done so that you could see how the titles would appear in a printed Word document. However, this is not what you see when you first open an ODS RTF file in Word.

When you first open an RTF document, you see it in normal view or page layout view (depending on your default settings). In this view, you might be surprised by how faint the titles and footnotes look. Instead of the bold fonts you were expecting, you see pale gray text. The RTF file from our previous examples is shown in Figure 4, where you can see that the titles are barely visible. This is because Word takes your titles and footnotes and turns them into page headings and footers, which are displayed in a grayed-out font on the screen.

Payroll·Report© The·MEANS·Procedure®©									
Analysis Variable∷ AnnualSalary Annual Salary⊙									
•Department=	Mean	Sum							
•Administration¤	51014	3571000×3							
•Manufacturing¤	51 507>	10456000×3							
•Marketing¤	53465	3849500×3							
•R&D¤	535878	2786500× ³							

Figure 4: Titles in Document Header

Because ODS creates RTF files that specify your titles and footnotes as headers and footers, they get converted to page headers and page footers in most word processors, and might not look correct on the screen. But rest assured, your document will print correctly.

If you would prefer that the titles and footnotes not be put in the header and footer, you can use the BODYTITLE option to force them into the body of the document. Here is the syntax:

ODS RTF FILE='filename' BODYTITLE;

Figure 5 shows the resulting RTF file. The only difference between the results shown here and the previous version shown in the figure above is that now the two headings are part of the main document and can be edited without switching to a different view.

Payroll·Report¶ The·MEANS·Procedure ¹ ¤									
Analysis Variable : AnnualSalary Annual Salary a									
Department	Mean	Sum							
Administration®	51014	3571000× [¤]							
Manufacturing¤	51 507)	10456000× [¤]							
Marketing¤	53465	3849500× [¤]							
R&D¤	53587;	2786500× [¤]							

Figure 5: Titles in Document Body

CONTROLLING PAGE ORIENTATION

When you have a table that is tall and skinny, it makes sense to print it in portrait layout. But when you have a table that is wider than it is tall, a landscape layout makes more sense. By default, if you don't change any options, ODS RTF creates portrait output.

Tone of content	N Obs	Variable	Label	N	Mean	Std Dev	Minimum	25th Pctl	Median	75th Pctl	Maximum
Negative	105	mentions compete	Number of references to company Number of references to competition	105 105	5.8 3.8	3.5 1.7	1.0 1.0	3.0 2.0	5.0 4.0	8.0 5.0	12.0 6.0
Neutral	111	mentions compete	Number of references to company Number of references to competition	111 111	6.2 3.2	3.7 1.7	1.0 1.0	3.0 2.0	6.0 3.0	10.0 5.0	12.0 6.0
Positive	114	mentions compete	Number of references to company Number of references to competition	114 114	7.0 3.2	3.0 1.7	1.0 1.0	5.0 2.0	7.0 3.0	9.0 4.0	12.0 6.0

The MEANS Procedure

Figure 6: Wide Table

If you have a wide table, like that shown in Figure 6, you might decide that you don't like the default portrait orientation. In this PROC MEANS analysis of the data set CLIPS, the variable labels for variables MENTIONS and COMPETE are too wide to fit their columns, so the text wraps and does not align with the statistics. This can be remedied by switching the orientation of the output to landscape. Here is the code that produces Figure 4.10:

```
ODS RTF FILE='CLIPSStats.rtf';
```

proc means data=clips n mean std min p25 median p75 max nway maxdec=1; class tone;

```
var mentions compete;
run;
ODS RTF CLOSE;
```

The syntax for changing the orientation is not part of ODS. Instead, you use the SAS system option ORIENTATION=. For example, the following statement placed before the ODF RTF statement above changes the orientation of the output to landscape and produces the output in Figure 7. Notice that the table is not much wider, but now there's room for the variable labels in their column.

OPTIONS ORIENTATION=LANDSCAPE;

The MEANS Procedure

Tone of content	N Obs	Variable	Label	N	Mean	Std Dev	Minimum	25th Pctl	Median	75th Pctl	Maximum
Negative	105	mentions compete	Number of references to company Number of references to competition	105 105	5.8 3.8	3.5 1.7	1.0 1.0	3.0 2.0	5.0 4.0	8.0 5.0	12.0 6.0
Neutral	111	mentions compete	Number of references to company Number of references to competition	111 111	6.2 3.2	3.7 1.7	1.0 1.0	3.0 2.0	6.0 3.0	10.0 5.0	12.0 6.0
Positive	114	mentions compete	Number of references to company Number of references to competition	114 114	7.0 3.2	3.0 1.7	1.0 1.0	5.0 2.0	7.0 3.0	9.0 4.0	12.0 6.0

Figure 7: Wide Table with Landscape Orientation

To switch the orientation back to the default, just issue the following command. Using this option makes it fairly simple to switch back and forth between orientations. It is also possible to switch orientations within a single RTF file.

OPTIONS ORIENTATION=PORTRAIT;

CONTROLLING PAGE BREAKS

When your RTF output produces a number of tables, ODS tries to insert page breaks appropriately. By default, it puts a page break between procedures and between BY groups. This can result in RTF files with a few tables spanning many pages with a lot of white space.

For example, if you run a PROC MEANS step followed by a PROC TABULATE step, you can end up with the results shown in Figure 8. The entire first page is devoted to the small table produced by the PROC MEANS step. The entire second page is devoted to the first BY group from the TABULATE procedure. These two pages are followed by four more pages, each displaying a single BY group from the PROC TABULATE step.



Figure 8: Output With Page Breaks

In HTML output, each of the procedures would just be displayed in order with no wasted space. You can achieve the same effect in RTF output by using the STARTPAGE= option. The syntax is as follows:

ODS RTF FILE='filename.rtf' STARTPAGE=NO;

This option setting tells ODS to not send any page breaks to the RTF output. The pagination of your results is handled by your word processor. For example, if this option is added to our sample code, the results now fit in the two pages shown in Figure 9.



Figure 9: Output Without Page Breaks

1

ADDING PAGE X OF Y

When you are producing multi-page reports, it's useful to be able to indicate how many pages the reader should expect. This avoids problems caused by someone not receiving all of the pages. If your SAS reports have page numbers in the format of "Page 1 of 3", "Page 2 of 3", etc., and a page gets lost in a fax transmission, the receiver will know to call you back and ask for the missing page.

Standard SAS page numbering just gives you a single page number when you turn on the following SAS system option setting:

OPTIONS NUMBER;

These page numbers are implemented in word processors by using a field that automatically computes the page number. This computation is useful because if you later edit your ODS RTF output, the page numbering will adjust automatically. In Word, you can use ALT-F9 to toggle the field so that it shows the code instead of the page number.

The same word processor fields are used to generate page numbers in the "Page X of Y" format. To ask ODS to create those fields, you need to turn off page numbering with the NONUMBER SAS system option, and you need to add an escape sequence to your TITLE statement that requests the "Page X of Y" text. The code centers the title text "Data Set CLIPS", and it right-justifies the "Page X of Y" text.

OPTIONS NONUMBER; ODS ESCAPECHAR='^'; title justify=center 'Data Set CLIPS' justify=right 'Page ^{pageof}';

For the RTF destination, SAS gives you a simple escape sequence to use to request the "X of Y" page numbers. The {pageof} escape sequence requests that ODS insert the word processor fields to generate the current page number and the total number of pages. You can see the results in Figure 10.

	Page 1 of 1						
01	one	one me	entions				
Neş	Nega	Negative	2	3	Newspaper	308	
₹eş	lega	Jegative	1	1	Radio	97	
Neş	Nega	Negative	5	2	Radio	88	
Nes	Nega	Negative	8	4	TV	327	

Figure 10: Page Numbers Generated by {pageof}

The field codes, which can be viewed by pressing ALT-F9 in Word, are shown in Figure 11.

Ē	L		Data S	et CLIP	8	Page	? { PAGE } of { NUMPAGES }
	Obs	tone	mentions	compete	media	words	
	1	Negative	2	3	Newspaper	308	
	2	Negative	1	1	Radio	97	
	3	Negative	5	2	Radio	88	
	4	M a matima	0	4	ти	227	

Figure 11: Field Codes Generated by {pageof}

CUSTOMIZING PAGE X OF Y

The method shown above has its limitations. The only page numbering format you can get with the {pageof} escape sequence is "X of Y". If you'd like something different, like "X/Y", you have to do that with different ODS escape character functions, as shown in the following specification. The {thispage} function returns the current page. The {lastpage} function returns the total number of pages in the document.

```
ODS ESCAPECHAR='^';
title justify=center 'Data Set CLIPS'
justify=right 'Page ^{thispage}/^{lastpage}';
```

These ODS ESCAPECHAR functions generate the actual RTF control strings for the page number in the desired format. This is a better alternative than using the equivalent RTF control strings, because the RTF control strings are very fussy about space, slashes, curly braces, and line breaks. If you mistype the RTF control strings, you might create an RTF file that cannot be opened.

ADDING STATIC DATES

Adding a date to a header or footer provides useful information about when a file was created or modified. By default, ODS inserts an RTF control string that corresponds to the Word date field in the page header of RTF output. This means that the date value is updated to the current date each time the file is opened.

Depending on why the date is needed, this might be what you want. However, for some reports, you want the date in the header to be the date the report was printed or delivered. For other reports, it might be preferable to have a fixed date that represents the date the report was generated.

With ODS RTF, you can create either static or dynamic dates. To generate a page header with a dynamic date that is updated every time the file is opened, you don't need to do anything. As long as the default SAS system option of DATE is in effect, you will get dynamic dates in your output.

To generate a page header with a static date based on the day the output was generated, all you have to do is add the SASDATE option to the ODS RTF statement. When you add this option, SAS inserts text for the current date when the program is run. Remember that the SAS date is actually the time and date that you started your SAS session. So if you keep your session open for days at a time, this date might not correspond to the actual date the program was run.

ODS RTF FILE='StaticDate.rtf' SASDATE;

ODS writes this date as a fixed value. If you toggle to the field codes view by pressing ALT-F9, the view of the value remains the same, and no field code is displayed.

ADDING IMAGES TO RTF OUTPUT

One of the quickest and easiest ways to dress up your RTF reports is to drop in a logo or graphic on each page. All you need to do is find a JPEG or PNG or other type of graphic and add a few special instructions to use your graphic. We've already seen the use of ODS ESCAPECHAR to control the page numbering. Now we're going to use it to insert a logo in the SAS title.

The PREIMAGE= attribute in the following TITLE statement specifies the filename of an image to use in the title. The filename specification must include path information that points to the location of the graphic file when the program is run. Because ODS embeds the image in the RTF file, you do not need to worry about keeping the graphics file along with the output file, as you would if you had sent the output to the HTML destination.

```
ODS NOPROCTITLE;
ODS RTF FILE='ToneFreqs.rtf';
ODS ESCAPECHAR='^';
title '^S={preimage="logo.jpg"}';
proc freq data=clips;
  tables tone;
run;
ODS RTF CLOSE;
```

A Header			ĹĴ	ſ		
			Fone of co	ntent		
				Cumulative	Cumulative	, n
	Tone	F Header and	Footer			▼
	Negative	Insei	t AutoText 🔻	* +	1 1 1	20 🖻 👺 🛃 🞜 🗐 🛛 🖸 Close
	Neutral	111	33.64	216	65.45	
	Positive	114	34.55	330	100.00	

Figure 12: Logo in Page Header

ODS places the graphic in the Header area, as shown in Figure 12. If you do not want the logo in the document header and instead want it placed in the document body, you can run the same code and add the BODYTITLE option to the ODS RTF statement. The logo would then appear in the body of the document.

The justification of the graphic is based on the system option CENTER. You can allow SAS to place your graphic based on the SAS system option setting, or you can specify a justification setting in your style. The following code adds the JUST= option, specifying a value of L for left. Two pages of output are shown in the figure below, which not only shows the left-justified logo, but also shows that the logo appears at the top of every page.

title '^S={just=l preimage="logo.jpg"}';



Figure 13: Left Justified Logos

ADDING TEXT

One of the reasons RTF is such a useful output destination is that we can cut and paste from the RTF document or we can add explanatory text to our RTF document to build our final report. But wouldn't it be nice if the entire report could be generated automatically?

To insert text into an RTF document, use the TEXT= option on an ODS statement to write to all open destinations, including RTF. The following statement inserts text into an open RTF destination file:

```
ODS TEXT='Text to insert';
```

You also need to add the STARTPAGE=OFF option to your ODS statement to prevent SAS from inserting page breaks between your table and text.

The output from the following program contains results from one ODS table from the PROC CORR step and adds text below the table. The ODS TEXT= statement follows the PROC CORR step. The text wraps across several lines, and only one statement is needed.

An escape character is defined at the beginning of the program. It is used to signal to ODS to apply left and right margins of one inch to the text. Figure 14 presents the output from this program.

```
ODS ESCAPECHAR='^';
ODS SELECT PearsonCorr;
ODS RTF FILE='AddText.rtf' STARTPAGE=OFF;
proc corr data=clips nosimple;
  var mentions;
  with compete;
run;
ODS TEXT='^S={LEFTMARGIN=1in RIGHTMARGIN=1in}
This text was inserted with the TEXT= option. It can be used to add a
detailed explanatory caption below a table or figure. This example uses
an inline formatting escape sequence to apply different styles here.
This text can also be formatted by using PROC TEMPLATE and the USERTEXT
style element';
ODS RTF CLOSE;
```

The CORR Proc	edure				
Pearson Correlation Coeff Prob > r under H0	Pearson Correlation Coefficients, N = 330 Prob > r under H0: Rho=0				
	Mentions				
Compete Number of references to competition	0.14996 0.0063				
This text was inserted with the TEXT= option. It can be use a table or figure. This example uses an inline formatting esc This text can also be formatted by using PROC TEMPLATE	i to add a detailed explanat ape sequence to apply diffe and the USERTEXT style	ory caption below rent styles here. element			

Figure 14: Output with Added Text

ADDING MORE COMPLEX TEXT

To generate a more complicated report, just add more procedures and more TEXT= options. The following code generates Figure 15 on the next page. Since we showed ODS TEXT= in the previous example, we show the other version of the TEXT= option in this example. This version of the TEXT= option writes only to the RTF destination, as opposed to the previous statement, which writes to ALL open destinations.

```
ODS ESCAPECHAR='^':
ODS SELECT PearsonCorr:
ODS RTF FILE='c:\temp\ReportWithText_4_38.rtf' STARTPAGE=OFF;
ODS RTF TEXT='^{style[font style=italic just=c fontweight=bold
fontsize=14pt]This bold, italicized text in 14pt introduces the table.
This introductory text will automatically wrap.}';
proc corr data=clips nosimple;
   var mentions;
  with compete;
run;
ODS RTF TEXT='^{newline 4}';
ODS RTF TEXT='^{style[font=(Arial)]Four lines appear before this
sentence (in Arial) and a bar chart follows:}';
proc gchart data=clips;
  vbar mentions / midpoints=1 to 12 by 1 inside=freq;
run;
quit;
ODS RTF TEXT='^{style[fontweight=bold textdecoration=underline]This
bold, underlined sentence concludes the report.}';
ODS RTF CLOSE;
```

If you wanted to customize the text, you could, for example, use macro variables to add the current date or insert custom text based on the results in the data. One thing to keep in mind is that as the text gets lengthy, you will start to get warning messages in the log about the string being longer than 512 characters. These messages can be ignored or turned off with the NOQUOTELENMAX option.



Figure 15: Output with Complex Added Text

ADDING HYPERLINKS TO RTF OUTPUT

Web pages aren't the only place you can use hyperlinks. You can put a link in a word processing document so that the reader can link to related information. With word processing documents frequently being shared electronically rather than on paper, this feature can be useful.

Adding a hyperlink is simple. You can add this link anywhere you have text in your output, such as labels, formatted values, titles, and footnotes. All you need to do is add the following escape sequence, where the escape character has been defined as the caret (^):

```
^S={URL="http://www.myweb.com"}
```

This code applies the style attribute URL= to the subsequent text, which specifies a destination for the hyperlink. Keep in mind that this original ODS ESCAPECHAR method applies the URL value to the whole cell and not just to a single word or sentence inside the cell, and it only works if the escape character (in this case, '^') is the first character in the text string.

The following example adds a hyperlink at the bottom of the output using the ODS RTF TEXT= statement. Rather than putting the hyperlink in a footnote, which would be at the bottom of the page, the TEXT= feature places the hyperlink in a text box right below the procedure output. Remember that if we used the ODS TEXT= form of the option, the text would be written to all open destinations.

```
ODS ESCAPECHAR='^';
ODS RTF FILE='Hyperlink.rtf';
proc freq data=clips;
  table compete;
run;
ODS RTF TEXT='^S={URL="http://www.myweb.com"}Source: www.myweb.com';
ODS RTF CLOSE;
```

The resulting RTF file is shown in Figure 16. You can see the text for the hyperlink below the table. If you move the mouse over the text, , you can see that it is a hyperlink. However, it's not obvious to readers that a hyperlink is there, unless they happen to move the mouse over the correct part of the page.

	Number of references to competition								
	Compete	Frequency	Percent	Cumulative Frequency	Cumulative Percent				
	1	58	17.58	58	17.58				
	2	64	19.39	122	36.97				
	3	50	15.15	172	52.12				
	4	57	17.27	229	69.39				
	5	52	15.76	281	85.15				
	б	49	14.85	330	100.00				
Source: www.myweb.com									



To solve this problem, you can provide a visual clue for the reader by changing the text color of the hyperlink to blue, the typical color used for hyperlinks. To do that, add a second escape sequence that sets the color of the text for the Web site address to blue, as in the following code:

```
ODS RTF TEXT='^S={URL="http://www.myweb.com"}Source:
^S={FOREGROUND=blue}www.myweb.com';
```

ADDING A TABLE OF CONTENTS

Starting with SAS®9, you can direct ODS to specify a table of contents in the RTF document that it creates. It does this by inserting control fields in your RTF document that your word processing software can use to build a table of contents. In SAS 8, you had to insert the RTF control fields yourself or build the table of contents manually from inside the word processing software.

In SAS 9.2, you add the CONTENTS= and TOC_DATA options to the ODS RTF statement to request a table of contents in the output.

ODS RTF FILE='PayrollTOC.rtf' CONTENTS=yes TOC_DATA;

This causes SAS to insert a table of contents as the first page of the report, as shown in Figure 17, and to insert control fields that support the table of contents in the pages of the report.

· · · · · · · · · · · · · · · · · · ·	α
Table of Contents	
Human Resources Report on Salary and Hours	2' <mark>n</mark>
Cross-tabular summary report	2
Table 1	2¶
Annual Salary Statistics for Administration	2¶
Annual Salary Statistics for Manufacturing	3¶
Annual Salary Statistics for Marketing	4¶
Annual Salary Statistics for R&D	5¶
Annual Salary Statistics in All Departments and Categories	6¶
Table 2	7
Hours per Week Statistics for Administration.	7¶
Hours per Week Statistics for Manufacturing	8¶
Hours per Week Statistics for Marketing	9¶
Hours per Week Statistics for R&D	10 Ï
Hours per Week Statistics for All Departments and Categories	11
Section Break (Next Page)	

Figure 17: ODS-Generated Table of Contents

The following program created this TOC. The ODS PROCLABEL statement replaces the procedure title with the text specified on the statement. Note that the labels for the page dimension in the PROC TABULATE step become references in the table of contents.

ODS RTF FILE='PayrollTOC.rtf' CONTENTS=yes TOC_DATA; ODS PROCLABEL 'Human Resources Report on Salary and Hours'; * proc tabulate code goes here ;

ODS RTF CLOSE;

When you run this code and open the resulting file, the table of contents does not automatically appear. The reason for this is that ODS does not determine pagination, but instead hands this task off to the word processor. ODS does not build a table of contents because it does not know the page numbers to insert into the table.

You must direct your word processing software to complete the table of contents. To do this after opening the RTF file, click the Show/Hide paragraph marker (¶) to turn on formatting markers if they are not already displayed. Next, position your cursor at the edge of the section break following the table of contents heading, right-click, and select Update Field. This causes Microsoft Word to build and insert the table of contents.

SPECIFYING METADATA

The ODS HTML statement options HEADTEXT= and METATEXT= can insert information into the <HEAD> section of your HTML file. This kind of information about your output file is called metadata. The ODS RTF statement also has options that control the insertion of metadata information into your RTF document. It is then up to your word processor to use the metadata information.

The kinds of metadata that you can tell ODS to store in your RTF document includes author name, document title, and operator name. This information is visible in Microsoft Word when you view the properties of the RTF document. To view the properties, open your document and select Properties from the File menu in Word 2003, or select Prepare ► Properties from the Office Button menu in Word 2007.

The following SAS 9.2 program specifies the three metadata options available on the ODS RTF statement: AUTHOR=, TITLE=, and OPERATOR=. The PROC MEANS step is identical to the one that produces Figure 4.1, and the report is not shown with this example.

```
ODS RTF FILE='Payroll_Meta.rtf'
AUTHOR="HR Report Designer"
TITLE="HR Standard Payroll Report"
OPERATOR="HR Assistant";
title 'Payroll Report';
proc means data=hr nonobs maxdec=0 mean sum;
class Department;
var AnnualSalary;
run;
ODS RTF CLOSE;
```

These are the generated RTF control strings which are put into your document:

{\info{\title HR Standard Payroll Report}{\author HR Report Designer}{\operator HR Assistant}

It is then up to the word processor (Word) to render the control strings appropriately. Figure 18 shows the properties of the file created by the preceding program. The Summary tab shows the title and author information. The Statistics tab shows that the value specified by the OPERATOR= option is inserted in the Last saved by field.

payroll_meta.rtf Properties	×	
General Summary Statistics Contents Custom	payroll_meta.rtf Properties	×
Title: HR Standard Payroll Report Subject:	General Summary Statistics Contents Custom Created: Monday, October 06, 2008 7:07:00 PM Modified: Monday, October 06, 2008 7:07:21 PM Accessed: Monday, October 06, 2008 7:07:48 PM	
Author: HR Report Designer	Princea:	-
Manager:	Last saved by: HR Assistant Revision number: 1	
Company:	Total editing time: 0 Minutes	

Figure 18: Properties Values set by ODS

ENHANCING RTF USING RTF CONTROL WORDS AND FIELD CODES

Another method, which is described below, is to place RTF control words and field codes directly in the TITLE or FOOTNOTE statements in the program that creates your RTF document. There are so many control words that we show only a selection here. At the end of the chapter, see "More Enhancement of RTF Output Using RTF Control Words" for another example of using RTF control words.

Style	RTF Control	Quoted String in TITLE or FOOTNOTE Statement
	Word(s)	
Bold	\b	"\b\iO Bold \bO \i Italicize"
Italicize	\i	
Bullet	\bullet	"\Bullet Bullet"
Double strike	\striked1	"\striked1 Double Strike \striked0 Regular Words"
Double	\dul	"\uldb Double Underline \uldb0"
underline		
Engrave	\impr	"\impr Engrave"
Subscript	\sub	"Subscript \sub 1"
Superscript	\super	"Superscript \super 2"
Outline	\outl	"\outl Outline"
Shadow	\shad	"\shad Shadow"
Foreground	\cfn	"\cf1 Foreground1 \cf2 Foreground2 \cf3
color		Foreground3"
Wave	\ulw	"\ulwave Wave underline";
underline		
Thick	\ulth	"\ulth Thick Underline"
underline		
Font size in	\fs24	"\fs48 Font Size in half points size 48 will be
half points		24pt in the document"

Table 1 lists selected RTF control words that you might find useful.

Table 1: RTF Control Words

The syntax starts with the escape character. Next, curly braces enclose the keyword raw, followed by the control words and text to be formatted. This entire specification is enclosed in double quotation marks. In the sample TITLE statement, all you have to do is substitute your control words and text where we show <string>. The RTF control words won't change unless Microsoft changes them.

title "^{raw <string>}";

Figures 19 and 20 on the next page show the result of using these control words in a series of ten TITLE and three FOOTNOTE statements.

Т

Bold Italicize	
Builet	
Double Strike Regular Words	
<u>Double Underline</u>	
Engrave	
Subscript 1	
Superscript ²	
Omline	
Shadow	
Foreground1 Foreground2 Foreground3	

Figure 19: Titles Using RTF Control Words

Wave underline <u>Thick Underline</u> Font Size in half points size 48 will be 24pt in the document

Figure 20: Footnotes Using RTF Control Words

CONCLUSION

These are just a sampling of ODS RTF techniques. Other related topics worth exploring: tagsets, style templates, ODS LAYOUT, and ODS DOCUMENT.

REFERENCES

Haworth, Lauren E., Zender, Cynthia L., and Burlew, Michele M. 2009. *Output Delivery System: The Basics and Beyond*. Cary, NC: SAS Institute Inc.

ACKNOWLEDGMENTS

The examples in this paper were co-written with Cynthia Zender, Curriculum Manager for the Report Writing and Output Delivery System curriculum at SAS, and Michele Burlew, owner of Episystems, Inc..

RECOMMENDED READING

Zender, Cynthia L. 2010. "SAS Style Templates: Always In Fashion." *Proceedings of the SAS Global Forum 2010 Conference*. Available at: <u>http://support.sas.com/resources/papers/proceedings10/033-2010.pdf</u>.

Hester, Wayne. 2006. "Teaching Your RTF Tagset to Do Clever Tricks." *Proceedings of the SAS Users Group International 31 Conference*. Available at: <u>http://www2.sas.com/proceedings/sugi31/067-31.pdf</u>.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at: info@laurenhaworth.com.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.